

Oplossing extra oefening 1

Opgave 1

a moet verwisseld worden met z. Ascii waarde van a => 97 en van z => 122
 $97+122=219$. Wanneer we een a hebben, vinden we de bijhorende ascii waarde van z door $219-97$ te doen. Wanneer we een z hebben, vinden we de bijhorende ascii waarde van a door $219-122$ te doen.

Dit gaat voor de andere letters op dezelfde manier.

```
>function spiegel(tekst) ...
    s:="";
    len:= strlen(tekst);
    i:=1;
    repeat while (i<=len)
        kar:= substring(tekst, i, i);
        testascii:=ascii(kar);
        if testascii<97 or testascii>122
            then s:="er zitten andere tekens dan kleine letters in de tekst";
                i:=len+1;
            else s:=s|char(219-testascii);
                i:=i+1;
            endif;
        end;
    return s;
endfunction

>gecodeerd:=spiegel("tinaelsjans")
    grmzvohqzmb
>spiegel(gecodeerd)
    tinaelsjans
```

Dit is een symmetrisch cryptosysteem. Elke letter a wordt door een z vervangen, door patroonherkenning kan je als de boodschap lang genoeg is de vercijferde boodschap gaan decoderen.

Opgave 2

Om een tekst te genereren, gaan we de asciiwaarden van de letters a tot z genereren. We moeten dus gehele getallen genereren ≥ 97 en ≤ 122 . We genereren gehele getallen ≥ 1 en ≤ 26 en tellen er daarna 96 bij op. Daarna converteren we de asciiwaarden naar de bijhorende karakters.

```
>function randomTekst(n) ...
    tekst="";
    for i=1 to n step 1
        random:=inrandom(26)+96;
        tekst:=tekst|char(random);
    end;
    return tekst;
endfunction
```

Uittesten van deze functies => getal genereren >10 en <20 of ≥ 11 en ≤ 19 . We genereren een geheel getal ≥ 1 en ≤ 9 en tellen er 10 bij op.

```
>n:=inrandom(9)+10
11
>randomTekst(11)
wdfmapcqfzv
>spiegel(%)
dwunzkxjuae
```

Opgave 3

```
>function caesar(v,sleutel) ...
    len:=length(v);
    i:=1;
    res:=[];
    repeat while (i<=len)
        if v[i]<0 or v[i] > 26
            then i:=len+2;
        else get:=mod(v[i]+sleutel, 26);
            res:=res|get;
            i:=i+1;
        endif;
    end;
    if i==len+2
        then return "fout in de invoer"
    else return res
    endif;
endfunction
```

- a) Genereren van tekst. Random getal ≥ 25 en <35 of geheel getal ≥ 25 en ≤ 34 . Geheel getal genereren ≥ 1 en ≤ 10 en hierbij 24 optellen.

```
>n:=inrandom(10)+24
27
```

```
>tekst:=randomTekst(n)
xhnilhhwhsflixfmcplsminnegw
```

b) Coderen

- Tekst spiegelen

```
>spiegel(tekst)
csmrossdshuorcunxkohnrmmvtd
```

- Tekst naar cijfer converteren

```
>naarCijfer(%)
[2, 18, 12, 17, 14, 18, 18, 3, 18, 7, 20, 14, 17, 2, 20,
13, 23, 10, 14, 7, 13, 17, 12, 12, 21, 19, 3]
```

- Caesarmethode toepassen met de juiste sleutel. We gebruiken hiervoor dezelfde sleutel als in oefening 1 in de cursus.

```
>sleutel:=ascii("E")
69
```

```
>caesar(%, sleutel)
[19, 9, 3, 8, 5, 9, 9, 20, 9, 24, 11, 5, 8, 19, 11, 4,
14, 1, 5, 24, 4, 8, 3, 3, 12, 10, 20]
```

- Cijfers omzetten naar tekst

```
>naarTekst(%)
tjdifjjujylfitleobfyeiddmku
```

c) decoderen

- Tekst naar cijfer converteren

```
>naarCijfer(%)
[19, 9, 3, 8, 5, 9, 9, 20, 9, 24, 11, 5, 8, 19, 11, 4,
14, 1, 5, 24, 4, 8, 3, 3, 12, 10, 20]
```

- Caesarmethode toepassen met de juiste sleutel

```
>caesar(%, -sleutel)
[2, 18, 12, 17, 14, 18, 18, 3, 18, 7, 20, 14, 17, 2, 20,
13, 23, 10, 14, 7, 13, 17, 12, 12, 21, 19, 3]
```

- Cijfers omzetten naar tekst

```
>naarTekst(%)
csmrossdshuorcunxkohnrmmvtd
```

- Tekst spiegelen

```
>spiegel(%)
xhnilhhwhsflixfmcplsminnegw
```